Step 1 : let’s define our prior as an exponential distribution

An exponential distribution is defined as
$f(\mu) = 2e^{-2\mu}$ for $\mu \geqslant 0$

based on the parameter values, we define the prior as a function of $\lambda = 2$

In[3]:=
```
prior[μ_] := PDF[ExponentialDistribution[2], μ]
```

let' s print its analytic form to see whether it works

In[4]:=
```
prior[μ]
```

Out[4]=
$$\begin{cases} 2\,e^{-2\mu} & \mu \geq 0 \\ 0 & \text{True} \end{cases}$$

let' s plot it as well

In[5]:=
```
Plot[prior[μ], {μ, 0, 3}, Filling → Axis]
```

Out[5]=


Step 2 : let’s define our likelihood function of each data point as a Normal Distribution

An Normal distribution with mean 2 and variance 1/2 is

$$f\,(x|\mu) = \frac{1}{\sqrt{\pi}} e^{-(x-\mu)^2}$$

Since variance is 1/2, the corresponding standard deviation is $\frac{1}{\sqrt{2}}$

In[6]:=
```
lf[x_] := PDF[NormalDistribution[μ, 1 / Sqrt[2]], x]
```

In[7]:=
```
lf[x]
```

Out[7]=

$$\frac{e^{-(x-\mu)^2}}{\sqrt{\pi}}$$

In[8]:=
```
Plot[Table[lf[x], {μ, {0, 1}}] // Evaluate,
 {x, -2, 3}, Filling → Axis]
```

Out[8]=



Step 3 : let's define our likelihood function of all data points as the product of Exponential PDFs

In[9]:=
```
likelihood[f_, data_] := Product[PDF[f, x], {x, data}]
```

In[10]:=
```
likelihood[NormalDistribution[μ, 1 / Sqrt[2]],
 {a1, a2, a3, a4, a5}]
```

Out[10]=
$$\frac{e^{-(-\mu+a_1)^2-(-\mu+a_2)^2-(-\mu+a_3)^2-(-\mu+a_4)^2-(-\mu+a_5)^2}}{\pi^{5/2}}$$

In[11]:=
```
l = Simplify[likelihood[
   NormalDistribution[μ, 1 / Sqrt[2]], {a1, a2, a3, a4, a5}]]
```

Out[11]=
$$\frac{e^{-(\mu-a_1)^2-(\mu-a_2)^2-(\mu-a_3)^2-(\mu-a_4)^2-(\mu-a_5)^2}}{\pi^{5/2}}$$

Step 4 : let's calculate our posterior using Bayes Rule

The Bayes Rule suggests that

$$\pi(\lambda|x) = \frac{L(x|\lambda)p(\lambda)}{\int L(x|\lambda)p(\lambda)\,d\lambda}$$

Since the denominator is just a constant, we just need to focus on the numerator calculation.

Let's also assume $\mu > 0$

In[12]:=
```
num = Assuming[μ > 0, Simplify[l * prior[μ]]]
```

Out[12]=
$$\frac{2\,e^{-2\mu-(\mu-a_1)^2-(\mu-a_2)^2-(\mu-a_3)^2-(\mu-a_4)^2-(\mu-a_5)^2}}{\pi^{5/2}}$$

We don't need to struggle with any constant terms;

hence, let's ignore $\pi^{5/2}$ and $\boxed{2}$

just focus on the exponent of the exponential function.

In[13]:=
```
s0 = Exponent[num, E]
```

Out[13]=
$$-2\,\mu-(\mu-a_1)^2-(\mu-a_2)^2-(\mu-a_3)^2-(\mu-a_4)^2-(\mu-a_5)^2$$

Again, we don’t need to struggle with any constant terms;
hence, let’s
1. expand the exponent

In[14]:=
```
s1 = Expand[s0]
```

Out[14]=
$$-2\,\mu-5\,\mu^2+2\,\mu\,a_1-a_1^2+2\,\mu\,a_2-a_2^2+2\,\mu\,a_3-a_3^2+2\,\mu\,a_4-a_4^2+2\,\mu\,a_5-a_5^2$$

2. remove all constant terms (terms without μ)

In[15]:=
```
s2 = Collect[s1, μ]
```

Out[15]=
$$-5\,\mu^2-a_1^2-a_2^2-a_3^2-a_4^2-a_5^2+\mu\,(-2+2\,a_1+2\,a_2+2\,a_3+2\,a_4+2\,a_5)$$

In[16]:=
```
s3 = Select[s2, MemberQ[#, μ | μ ^ _] &]
```

Out[16]=
$$-5\,\mu^2+\mu\,(-2+2\,a_1+2\,a_2+2\,a_3+2\,a_4+2\,a_5)$$

Hence, all the manipulations above can be summarized into the following operation

In[17]:=
```
result = Select[Collect[Expand[Exponent[num, E]], μ],
  MemberQ[#, μ | μ ^ _] &]
```

Out[17]=
$$-5\,\mu^2+\mu\,(-2+2\,a_1+2\,a_2+2\,a_3+2\,a_4+2\,a_5)$$

put the simplified exponent back to an expoential function, we have

In[18]:=
```
Exp[result]
```

Out[18]=
$$e^{-5\,\mu^2+\mu\,(-2+2\,a_1+2\,a_2+2\,a_3+2\,a_4+2\,a_5)}$$

It looks like part of the Normal PDF

$$f\,(x|\mu,\sigma)=\frac{1}{\sqrt{2\pi}\sigma}e^{-\frac{(x-\mu)^2}{2\sigma^2}}\ \text{ where } \sigma\geqslant 0$$

Let’s vary our sample points (add one extra sample points) to find the pattern of parameters.

In[19]:=
```
l = Simplify[likelihood[NormalDistribution[μ, 1/Sqrt[2]],
    {a1, a2, a3, a4, a5, a6}]];
num = Assuming[μ > 0, Simplify[l * prior[μ]]];
result = Select[Collect[Expand[Exponent[num, E]], μ],
  MemberQ[#, μ | μ^_] &]
```

Out[21]= $-6\,\mu^2+\mu\,(-2+2\,a_1+2\,a_2+2\,a_3+2\,a_4+2\,a_5+2\,a_6)$

In[22]:=
```
l = Simplify[likelihood[
    NormalDistribution[μ, 1/Sqrt[2]], {a1, a2, a3, a4}]];
num = Assuming[μ > 0, Simplify[l * prior[μ]]];
result = Select[Collect[Expand[Exponent[num, E]], μ],
  MemberQ[#, μ | μ^_] &]
```

Out[24]= $-4\,\mu^2+\mu\,(-2+2\,a_1+2\,a_2+2\,a_3+2\,a_4)$

Hence, we know that the coefficient of μ^2 is the sample size n.

The final step is to rearrange $e^{(2\,a_1+2\,a_2+2\,a_3+2\,a_4+2\,a_5-2)\,\mu-5\,\mu^2}$ into a Normal PDF

We define $A := \frac{1}{n}\,(a_1+a_2+a_3+a_4+a_5)$ and replace the coefficient of μ^2 as sample size n.

In[25]:=
```
p0 = E^(-n*(μ^2 - 2 μ (A - 1/n)))
```

Out[25]= $e^{-n\left(-2\left(A-\frac{1}{n}\right)\mu+\mu^2\right)}$

In[26]:=
```
p1 = E^(-n*(μ^2 - 2 μ (A - 1/n) + (A - 1/n)^2))
```

Out[26]= $e^{-n\left(\left(A-\frac{1}{n}\right)^2-2\left(A-\frac{1}{n}\right)\mu+\mu^2\right)}$

In[27]:=
```
p2 = E^(-n*(μ - (A - 1/n))^2)
```

Out[27]= $e^{-n\left(-A+\frac{1}{n}+\mu\right)^2}$

Now you can see the posterior is

$$\exp\left\{-\frac{\left[\mu-\left(A-\frac{1}{n}\right)\right]^2}{2\left(1\Big/\sqrt{2\,n}\right)^2}\right\}=\exp\left\{-n\,[\mu-(A-1/n)]^2\right\}$$

which is a Normal distribution